■ 管路I,IIを持つ複合管のインピーダンスの計算

管路Iの入口出口の音圧、体積速度をp1,u1,p2,u2とし、
管路IIをq1,v1,q2,v2とするとまず次の関係式があるはず
ここに、m11...,n11...は伝達行列の成分である。

```
eq1 = {p1, u1} == {{m11, m12}, {m21, m22}}.{p2, u2}
```

```
{p1, u1} == {m11 p2 + m12 u2, m21 p2 + m22 u2}
```

```
eq2 = {q1, v1} == {{n11, n12}, {n21, n22}}.{q2, v2}
```

```
{q1, v1} == {n11 q2 + n12 v2, n21 q2 + n22 v2}
```

これらを解いて

```
s1 = Solve[eq1, {u1, u2}]
```

$$\left\{\left\{u1 \to -\left(\frac{-(m22\ p1) - m12\ m21\ p2 + m11\ m22\ p2}{m12}\right),\ u2 \to -\left(\frac{-p1 + m11\ p2}{m12}\right)\right\}\right\}$$

```
s2 = Solve[eq2, {v1, v2}]
```

$$\left\{\left\{v1 \to -\left(\frac{-(n22\ q1) - n12\ n21\ q2 + n11\ n22\ q2}{n12}\right),\ v2 \to -\left(\frac{-q1 + n11\ q2}{n12}\right)\right\}\right\}$$

入口出口がつながった複合管では各口の音圧は連続である

```
ss = {p1 → P1, q1 → P1, p2 → P2, q2 → P2}
```

```
{p1 -> P1, q1 -> P1, p2 -> P2, q2 -> P2}
```

入口出口の全体積速度をU1,U2とすると体積速度の関係式は

```
eq3 = U1 == u1 + v1 /. s1 /. s2 /. ss
```

$$\left\{\left\{U1 == -\left(\frac{-(m22\ P1) - m12\ m21\ P2 + m11\ m22\ P2}{m12}\right) - \frac{-(n22\ P1) - n12\ n21\ P2 + n11\ n22\ P2}{n12}\right\}\right\}$$

```
eq4 = U2 == u2 + v2 /. s1 /. s2 /. ss
```

$$\left\{\left\{U2 == -\left(\frac{-P1 + m11\ P2}{m12}\right) - \frac{-P1 + n11\ P2}{n12}\right\}\right\}$$

2つの式をP1,U1の組で表す

```
eq5 = {eq3, eq4} // Flatten
```

$$\{U1 == -\left(\frac{-(m22\ P1) - m12\ m21\ P2 + m11\ m22\ P2}{m12}\right) - \frac{-(n22\ P1) - n12\ n21\ P2 + n11\ n22\ P2}{n12},\ U2 == -\left(\frac{-P1 + m11\ P2}{m12}\right) - \frac{-P1 + n11\ P2}{n12}\}$$

```
s5 = Solve[eq5, {P1, U1}] // First
```

$$\{U1 \to -((-(m12\ m21\ P2) + m11\ m22\ P2 - m22\ n11\ P2 - m21\ n12\ P2 - m12\ n21\ P2 - n12\ n21\ P2 - m11\ n22\ P2 + n11\ n22\ P2 - m22\ n12\ U2 - m12\ n22\ U2)\ /\ (m12 + n12))\backslash,\ P1 \to -\left(\frac{-(m12\ n11\ P2) - m11\ n12\ P2 - m12\ n12\ U2}{m12 + n12}\right)\}$$

複合管の入力インピーダンスZ1,出口をZ2

```
Z1 = P1 / U1 /. s5 /. {P2 → Z2 U2} // FullSimplify
```

$$\frac{(m11\ n12\ Z2 + m12\ (n12 + n11\ Z2))}{(m22\ n12 + m12\ n22 + ((m12 + n12)\ (m21 + n21) - (m11 - n11)\ (m22 - n22))\ Z2)}$$

```
% // TraditionalForm
```

$$\frac{m11\,n12\,Z2 + m12\,(n12 + n11\,Z2)}{m22\,n12 + m12\,n22 + ((m12 + n12)\,(m21 + n21) - (m11 - n11)\,(m22 - n22))\,Z2}$$

```
MapAt[Collect[#, Z2] &, %, 2] // TraditionalForm
```

$$\frac{m12\,n12 + (m12\,n11 + m11\,n12)\,Z2}{m22\,n12 + m12\,n22 + ((m12 + n12)\,(m21 + n21) - (m11 - n11)\,(m22 - n22))\,Z2}$$

```
% // CForm
```

```
(m12*n12 + (m12*n11 + m11*n12)*Z2)/
   (m22*n12 + m12*n22 + ((m12 + n12)*(m21 + n21) -
        (m11 - n11)*(m22 - n22))*Z2)
```

特殊な例

```
Z1 /. m12 → 0
```

$$\frac{m11\ n12\ Z2}{m22\ n12 + (n12\ (m21 + n21) - (m11 - n11)\ (m22 - n22))\ Z2}$$

```
Z1 /. n12 → 0
```

$$\frac{m12\ n11\ Z2}{m12\ n22 + (m12\ (m21 + n21) - (m11 - n11)\ (m22 - n22))\ Z2}$$

```
Z1 /. {m12 → 0, n12 → 0}
```

```
0
```

```
Z1 /. {m11 → n11}
```

$$\frac{n11\ n12\ Z2 + m12\ (n12 + n11\ Z2)}{m22\ n12 + m12\ n22 + (m12 + n12)\ (m21 + n21)\ Z2}$$

```
Z1 /. {m11 → n11, m22 → n22}
```

$$\frac{n11\ n12\ Z2 + m12\ (n12 + n11\ Z2)}{m12\ n22 + n12\ n22 + (m12 + n12)\ (m21 + n21)\ Z2}$$

```
Z1 /. {m11 → n11, m22 → n22, m12 → n12, m21 → n21} // FullSimplify
```

$$\frac{n12 + 2\ n11\ Z2}{2\ n22 + 4\ n21\ Z2}$$